
1. Derive a program for the computation of square root.

con N : Int {0 ⩽ N}
var x : Int

squareroot
{x2 ⩽ N < (x + 1)2} .

Solution: Try using x2 ⩽ N as the invariant and ¬(N < (x + 1)2) as the guard. The program:

con N : Int {0 ⩽ N}
var x : Int

x := 0 -- Pf0
{x2 ⩽ N , bnd : N − x} -- Pf1
do (¬ (N < (x + 1)2)) →
x := x + 1 -- Pf2

od
{x2 ⩽ N < (x + 1)2} -- Pf3

Pf0.

(x2 ⩽ N)[x\0]
≡ 02 ⩽ N
≡ 0 ⩽ N .

Pf1. To show that the bound is non-negative:

0 ⩽ N − x
≡ x ⩽ N
⇐ { x ⩽ x2 for integer x }

x2 ⩽ N
⇐ x2 ⩽ N ∧ ¬ (N < (x + 1)2) .

To show that the bound decreases:

(N − x < C)[x\x + 1]
≡ N − x − 1< C
⇐ N − x = C
⇐ N − x = C ∧ x2 ⩽ N ∧ ¬ (N < (x + 1)2) .

Note: what would happen had we chosen N − x2 as the bound?

Pf2.
(x2 ⩽ N)[x\x + 1]

≡ (x + 1)2 ⩽ N
⇐ x2 ⩽ N ∧ ¬ (N < (x + 1)2) .

Pf3. Certainly,

x2 ⩽ N ∧ ¬ (¬ (N < (x + 1)2))
≡ x2 ⩽ N < (x + 1)2 .

2. For each implication below, find a substitution (on variables) such that the implication holds. Note:

• Names starting with small letters (x , a, b, etc) are variables, while A, B, and C are constants. E denotes an
expression.

• We assume that all variables and constants are Int .

• For some questions, there could be more than one substitutions that work.

(a) (x = 2× E)[?\?] ⇐ x = E , where x does not occur free in E .

(b) (x = 2× E + A)[?\?] ⇐ x = E , where x does not occur free in E .

(c) (x = f E)[?\?] ⇐ x = E , for some function f . Again, x does not occur free in E .

(d) (x = A)[?\?] ⇐ x = 2× A + B.

(e) (A = 2 × b × x + c)[?\?] ⇐ A = b × x + c ∧ …. You may need to discover an additional condition in … to
make the implication valid.

(f) (A = B× x + B + C)[?\?] ⇐ A = B× x + C.

(g) (A = B × x / 2 + 2 × C)[?\?] ⇐ A = B × x + C ∧ …. You will need a side condition. Note that (×) and (/)
are left-associative. That is, B× X / C is interpreted as (B× X ) / C.

Solution:

(a) [x\2× x].

(b) [x\2× x + A].

(c) [x\f x].

(d) One may choose [x\((x − B) / 2)]. That is,

(x = A)[x\((x − B) / 2)]
≡ (x − B) / 2 = A
≡ x = 2× A + B .

Another possibility could be: [x\(x − A)− B]:

(x = A)[x\(x − A)− B]
≡ (x − A)− B = A
≡ x = 2× A + B .

(e) One may choose [x\(x / 2)] with an additional condition even x :

(A = 2× b × x + c)[x\x / 2]
≡ A = 2× b × (x / 2) + c
⇐ A = b × x + c ∧ even x .

Note that, since x :Int and (/) is integral division, we need even x to guarantee that 2×b×(x/2) = b×x .

One could also choose [b\(b / 2)] with an additional condition even b, or [c\(c − b × x)].

(f) [x\x − 1].

(g) [x\(2× x − 2× C / B)], with side condition 2× C ‘mod ‘ B = 0, that is B divides 2× C:

(A = B× x / 2 + 2× C)[x\(2× x − 2× C / B)]
≡ A = B× (2× x − 2× C / B) / 2 + 2× C
≡ A = (B× 2× x − B× 2× C / B) / 2 + 2× C
⇐ { B× X / B = X if B divides X }

A = (B× 2× x − 2× C) / 2 + 2× C ∧ 2× C ‘mod ‘ B = 0
≡ A = B× x − C + 2× C ∧ 2× C ‘mod ‘ B = 0
≡ A = B× x + C ∧ 2× C ‘mod ‘ B = 0 .

3. The Zune problem. Let D be the number of days since 1st January 1980. What is the current year? Assume
that there exists a function daysInYear : Int → Int such that daysInYear i, with i ⩾ 1980, yields the number of
days in year i, which is always a positive number. Derive a program having two variables y and d such that,
upon termination, y is the current year, and d is the number of days since the beginning of this year.

(a) How would you specify the problem? The specification may look like:

con D : Int {0 ⩽ D}
var y , d : Int

zune
{???}

What would you put as the postcondition? In this postcondition, is 1st January 1980 day 0 or 1?

Solution: One of the possibilities is

⟨Σi : 1980 ⩽ i < y : daysInYear i⟩ + d = D ∧ 0 ⩽ d < daysInYear y .

This specification implies that 1st January 1980 is day 0 and, days in year i are counted as 0, 1 …
daysInYear i − 1.

(b) Derive the program.

Solution: We choose ⟨Σi : 1980 ⩽ i < y : daysInYear i⟩ + d = D ∧ 0 ⩽ d as the loop invariant, and
¬ (d < daysInYear y) as guard. During the development we will see that we need 1980 ⩽ y in the
invariant, to allow splitting. The resulting program is:

con D : Int {0 ⩽ D}
var y , d : Int

y , d := 1980,D -- Pf0
{⟨Σi : 1980 ⩽ i < y : daysInYear i⟩ + d = D ∧ 1980 ⩽ y ∧ 0 ⩽ d , bnd : d}
do d ⩾ daysInYear y → -- Pf1
d := d − daysInYear y -- Pf2
y := y + 1

od
{⟨Σi : 1980 ⩽ i < y : daysInYear i⟩ + d = D ∧ 0 ⩽ d < daysInYear y} -- Pf3

Pf0.

(⟨Σi : 1980 ⩽ i < y : daysInYear i⟩ + d = D ∧ 1980 ⩽ y ∧ 0 ⩽ d)[y , d\1980,D]
≡ ⟨Σi : 1980 ⩽ i < 1980 : daysInYear i⟩ + D = D ∧ 1980 ⩽ 1980 ∧ 0 ⩽ D
≡ 0 + D = D ∧ 0 ⩽ D
≡ 0 ⩽ D .

Pf1. That 0 ⩽ d follows from the loop invariant. To show that d decreases, we need to know that
daysInYear y is always positive:

((d < C)[y\y + 1])[d\d − daysInYear y]
≡ d − daysInYear y < C
⇐ { daysInYear y positive }

d = C
⇐ ⟨Σi : 1980 ⩽ i < y : daysInYear i⟩ + d = D ∧ 1980 ⩽ y ∧ 0 ⩽ d ∧ d ⩾ daysInYear y ∧ d = C .

Pf2. Assuming 1980 ⩽ y , consider

⟨Σi : 1980 ⩽ i < y : daysInYear i⟩[y\y + 1]
= ⟨Σi : 1980 ⩽ i < y + 1 : daysInYear i⟩
= { since 1980 ⩽ y , splitting off i = y }
⟨Σi : 1980 ⩽ i < y : daysInYear i⟩ + daysInYear y .

Therefore,

((⟨Σi : 1980 ⩽ i < y : daysInYear i⟩ + d = D ∧
1980 ⩽ y ∧ 0 ⩽ d)[y\y + 1])[d\d − daysInYear y]

≡ ⟨Σi : 1980 ⩽ i < y + 1 : daysInYear i⟩ + (d − daysInYear y) = D ∧
1980 ⩽ y + 1 ∧ 0 ⩽ d − daysInYear y

⇐ { calculation above, 1980 ⩽ y + 1 ⇐ 1980 ⩽ y }
⟨Σi : 1980 ⩽ i < y : daysInYear i⟩ + daysInYear y + (d − daysInYear y) = D ∧

1980 ⩽ y ∧ d ⩾ daysInYear y
⇐ ⟨Σi : 1980 ⩽ i < y : daysInYear i⟩ + d = D ∧ 1980 ⩽ y ∧ 0 ⩽ d ∧ d ⩾ daysInYear y .

Pf3. Certainly,

⟨Σi : 1980 ⩽ i < y : daysInYear i⟩ + d = D ∧ 1980 ⩽ y ∧ 0 ⩽ d ∧
¬ (d ⩾ daysInYear y) ⇒
⟨Σi : 1980 ⩽ i < y : daysInYear i⟩ + d = D ∧ 0 ⩽ d < daysInYear y .

4. Assuming that −∞ is the identity element of (↑). Derive a solution for:

con N : Int {N ⩾ 0}
con A : array [0..N) of Int
var r : Int
S
{r = ⟨↑ i : 0 ⩽ i < N : A[i]⟩} .

Solution:

con N : Int {N ⩾ 0}
con A : array [0..N) of Int
var r , n : Int
r , n :=−∞, 0 -- Pf0
{r = ⟨↑ i : 0 ⩽ i < n : A[i]⟩ ∧ 0 ⩽ n ⩽ N , bnd : N − n}
do n ̸= N → -- Pf1
r := r ↑ A [n] -- Pf2
n := n + 1

od
{r = ⟨↑ i : 0 ⩽ i < N : A[i]⟩} -- Pf3

Pf0.

(r = ⟨↑ i : 0 ⩽ i < n : A [ i ]⟩ ∧ 0 ⩽ n ⩽ N)[r , n\−∞, 0]
≡ −∞ = ⟨↑ i : 0 ⩽ i < 0 : A [ i ]⟩ ∧ 0 ⩽ 0 ⩽ N
≡ 0 ⩽ N .

Pf1. Apparently, 0 ⩽ n ⩽ N ⇒ N − n ⩾ 0, and

((N − n< C)[n\n + 1])[r\r ↑ A[n]]
≡ N − (n + 1)< C
⇐ N − n = C .

Pf2. We reason:

((r = ⟨↑ i : 0 ⩽ i < n : A[i]⟩ ∧ 0 ⩽ n ⩽ N)[n\n + 1])[r\r ↑ A[n]]
≡ r ↑ A[n] = ⟨↑ i : 0 ⩽ i < n + 1 : A[i]⟩ ∧ 0 ⩽ n + 1 ⩽ N
⇐ { assuming 0 ⩽ n< N , split off i = n }

r ↑ A[n] = ⟨↑ i : 0 ⩽ i < n : A[i]⟩ ↑ A[n] ∧ 0 ⩽ n< N
⇐ r = ⟨↑ i : 0 ⩽ i < n : A[i]⟩ ∧ 0 ⩽ n ⩽ N ∧ n ̸= N .

Pf3. It is immediate that

r = ⟨ ↑ i : 0 ⩽ i < n : A[i] ⟩ ∧ 0 ⩽ n ⩽ N ∧ n = N

⇒ r = ⟨ ↑ i : 0 ⩽ i < N : A[i] ⟩ .

5. Derive a solution for:

con N ,X : Int {0 ⩽ N}
con A : array [0..N) of Int
var r : Int
S
{r = ⟨Σi : 0 ⩽ i < N : A [ i ]× X i⟩} .

Solution: For efficiency, add a variable x and use the invariant:

r = ⟨Σi : 0 ⩽ i < n : A [ i ]× X i⟩ ∧ x = Xn ∧ 0 ⩽ n ⩽ N .

Denote it by P . The program:

con N ,X : Int {0 ⩽ N}
con A : array [0..N) of Int
var r , x , n : Int
r , x , n := 0, 1, 0 -- Pf0
{P , bnd : N − n}
do n ̸= N → -- Pf1
r , x := r + A [n]× x , x × X -- Pf2
n := n + 1

od
{r = ⟨Σi : 0 ⩽ i < N : A [ i ]× X i⟩} -- Pf3

Pf0.

P[r , x , n\0, 1, 0]
≡ 0 = ⟨Σi : 0 ⩽ i < 0 : A [ i ]× X i⟩ ∧ 1 = X 0 ∧ 0 ⩽ 0 ⩽ N
⇐ 0 ⩽ N .

Pf1. Apparently, 0 ⩽ n ⩽ N ⇒ N − n ⩾ 0, and

((N − n< C)[n\n + 1])[r , x\r + A[n], x × X]
≡ N − (n + 1)< C
⇐ N − n = C .

Pf2. We reason:

((r = ⟨Σi : 0 ⩽ i < n : A[i]× X i⟩ ∧ x = Xn ∧ 0 ⩽ n ⩽ N)[n\n + 1])[r , x\r + A[n]× x , x × X]
≡ r + A[n]× x = ⟨Σi : 0 ⩽ i < n + 1 : A[i]× X i⟩ ∧ x × X = Xn+1 ∧ 0 ⩽ n + 1 ⩽ N
⇐ { assuming 0 ⩽ n< N , split off i = n }

r + A[n]× x = ⟨Σi : 0 ⩽ i < n : A[i]× X i⟩ + A[n]× xn ∧ x × X = Xn+1 ∧ 0 ⩽ n< N
⇐ r = ⟨Σi : 0 ⩽ i < n : A[i]× X i⟩ ∧ x = Xn ∧ 0 ⩽ n ⩽ N ∧ n ̸= N .

Pf3. It is immediate that

r = ⟨Σi : 0 ⩽ i < n : A[i]× X i ⟩ ∧ x = Xn ∧ 0 ⩽ n ⩽ N ∧ n = N

⇒ r = ⟨Σi : 0 ⩽ i < N : A[i]× X i ⟩.

Another possibility, however, is to define for 0 ⩽ n ⩽ N :

k n = ⟨Σi : n ⩽ i < N : A[i]× X i−n ⟩,

use the invariant r = k n ∧ 0 ⩽ n ⩽ N , and decrement n in the loop.

